


Labeled Property Graph

- Enthält Knoten (Nodes) und Beziehungen (Relationships)
- Knoten haben Eigenschaften (Properties, Key Value Pairs)
- Knoten sind markiert (labeled, one ore more Labels)
- Beziehungen haben Typen, sind markiert, gerichtet und haben einen Start- und Endknoten
- Beziehungen haben Eigenschaften




Übernommen aus:

Ian Robinson, Jim Webber & Emil Eifrem
Graph Databases, Second Edition,
O'Reilly, 2015


Die folgenden Folien basieren auf Material von Neo4j, welches unter folgender Lizenz zur Verfügung gestellt wird:

<https://neo4j.com/docs/license/>



MATCH (p:Person)

RETURN p



```
{  
  "identity": 41,  
  "labels": [  
    "Person"  
  ],  
  "properties": {  
    "name": "Val Kilmer",  
    "born": 1959  
  }  
}
```

```
MATCH (p:Person {born: 1970})  
RETURN p
```




```
MATCH (m:Movie {released: 2003, tagline: 'Free your mind'})  
RETURN m
```



```
MATCH (p:Person {born: 1965})  
RETURN p.name AS name, p.born AS birthyear  
ORDER BY p.name
```

"name"	"birthyear"
"John C. Reilly"	1965
"Lana Wachowski"	1965
"Tom Tykwer"	1965

```
MATCH (p:Person)-[rel:ACTED_IN]->(m:Movie {title: 'The Matrix'})  
RETURN p, m
```



```
MATCH (p:Person)-->(m:Movie {title: 'The Matrix'})  
RETURN p, m
```




```
MATCH (p:Person {name: 'Tom Hanks'})-[r:ACTED_IN|DIRECTED]->(m:Movie)
RETURN p.name, m.title, type(r)
ORDER BY p.name
```


"p.name"	"m.title"	"type(r)"
"Tom Hanks"	"You've Got Mail"	"ACTED_IN"
"Tom Hanks"	"Apollo 13"	"ACTED_IN"
"Tom Hanks"	"Joe Versus the Volcano"	"ACTED_IN"
"Tom Hanks"	"That Thing You Do"	"ACTED_IN"
"Tom Hanks"	"Cloud Atlas"	"ACTED_IN"
"Tom Hanks"	"The Da Vinci Code"	"ACTED_IN"
"Tom Hanks"	"Sleepless in Seattle"	"ACTED_IN"
"Tom Hanks"	"A League of Their Own"	"ACTED_IN"
"Tom Hanks"	"The Green Mile"	"ACTED_IN"
"Tom Hanks"	"Charlie Wilson's War"	"ACTED_IN"

```
MATCH (p:Person)-[:REVIEWED {rating: 65}]->(:Movie {title: 'The Da Vinci Code'})  
RETURN p.name
```

"p.name"
"James Thompson"

MATCH path = (jt:Person {name:'James Thompson'})-[:FOLLOWS*]-(:Person {name:'Paul Blythe'})

RETURN path



```


{
  "start": {
    "identity": 162,
    "labels": [
      "Person"
    ],
    "properties": {
      "name": "James Thompson"
    }
  },
  "end": {
    "identity": 159,
    "labels": [
      "Person"
    ],
    "properties": {
      "name": "Paul Blythe"
    }
  },
  "segments": [
    ...
  ]
}
  
```

Ein einzelner Pfad

```


{
  "start": {
    "identity": 162,
    "labels": [
      "Person"
    ],
    "properties": {
      "name": "James Thompson"
    }
  },
  "relationship": {
    "identity": 239,
    "start": 162,
    "end": 161,
    "type": "FOLLOWS",
    "properties": {}
  },
  "end": {
    "identity": 161,
    "labels": [
      "Person"
    ],
    "properties": {
      "name": "Jessica Thompson"
    }
  }
},
...
  
```

```
MATCH path = (:Person)-[:ACTED_IN]->(:Movie)<-[:DIRECTED]-(:Person {name:'Ron Howard'})
RETURN path
```



Viele Pfade

```
MATCH (p:Person)-[:ACTED_IN]->(m:Movie)
WHERE m.released = 2008 AND p.born > 1960
RETURN p, m
```




```
MATCH (p:Person)-[:ACTED_IN]->(m:Movie)
WHERE p.name='Jack Nicholson' AND m.tagline IS NOT NULL
RETURN m.title
```

"m.title"
"A Few Good Men"
"As Good as It Gets"
"Hoffa"
"One Flew Over the Cuckoo's Nest"

```
MATCH (p:Person)
WHERE p.name =~'Tom.*'
RETURN p.name
```

"p.name"
"Tom Cruise"
"Tom Skerritt"
"Tom Hanks"
"Tom Tykwer"

```
MATCH (gene:Person)-[:ACTED_IN]->(m:Movie)<-[:ACTED_IN]-(other:Person)
WHERE gene.name = 'Gene Hackman'
AND exists((other)-[:DIRECTED]->(m))
RETURN gene, other, m
```




```
MATCH (p:Person)-[r:ACTED_IN]->(m:Movie)
WHERE 'Neo' IN r.roles AND m.title = 'The Matrix'
RETURN p.name
```

"p.name"
"Keanu Reeves"

```
MATCH (p:Person)-[r:ACTED_IN]->(m:Movie)
WHERE size(r.roles)=3
RETURN p.name, m.title, r.roles
```

"p.name"	"m.title"	"r.roles"
"Meg Ryan"	"Joe Versus the Volcano"	["DeDe", "Angelica Graynamore", "Patricia Graynamore"]
"Jim Broadbent"	"Cloud Atlas"	["Vyvyan Ayrns", "Captain Molyneux", "Timothy Cavendish"]

MATCH

```
(meg:Person)-[:ACTED_IN]->(m:Movie)<-[:DIRECTED]-(d:Person),
(other:Person)-[:ACTED_IN]->(m)
```


WHERE meg.name = 'Meg Ryan'

RETURN m.title as movie, d.name AS director, other.name AS `co-actors`


"movie"	"director"	"co-actors"
"Sleepless in Seattle"	"Nora Ephron"	"Victor Garber"
"Sleepless in Seattle"	"Nora Ephron"	"Tom Hanks"
"Sleepless in Seattle"	"Nora Ephron"	"Bill Pullman"
"Sleepless in Seattle"	"Nora Ephron"	"Rita Wilson"
"Sleepless in Seattle"	"Nora Ephron"	"Rosie O'Donnell"
"You've Got Mail"	"Nora Ephron"	"Tom Hanks"
"You've Got Mail"	"Nora Ephron"	"Parker Posey"
"You've Got Mail"	"Nora Ephron"	"Greg Kinnear"
"You've Got Mail"	"Nora Ephron"	"Steve Zahn"
"You've Got Mail"	"Nora Ephron"	"Dave Chappelle"

```
MATCH p = ((m1:Movie)-[:ACTED_IN*..6]-(m2:Movie))
WHERE m1.title = 'A Few Good Men' AND m2.title = 'The Matrix'
RETURN p
```

*	beliebig lang
*2..4	zwischen 2 und 4
*..6	bis 6
*3..	3 und länger



```
MATCH p = shortestPath((m1:Movie)-[:ACTED_IN*]-(m2:Movie))  
WHERE m1.title = 'A Few Good Men' AND  
m2.title = 'The Matrix'  
RETURN p
```



MATCH

```
(a:Person)-[:ACTED_IN]->(m:Movie)<-[:DIRECTED]-(d:Person)
RETURN a.name, d.name, count(m)
ORDER BY a.name
```

"a.name"	"d.name"	"count(m)"
"Aaron Sorkin"	"Rob Reiner"	1
"Al Pacino"	"Taylor Hackford"	1
"Annabella Sciorra"	"Vincent Ward"	1
"Anthony Edwards"	"Tony Scott"	1
"Audrey Tautou"	"Ron Howard"	1
"Ben Miles"	"James Marshall"	2
"Ben Miles"	"Andy Wachowski"	1
"Ben Miles"	"Lana Wachowski"	1
"Bill Dauterive"	"Don Edwards"	1

MATCH

```
(p:Person)-[:ACTED_IN]->(m:Movie)
WHERE p.name = 'Tom Cruise'
RETURN collect(m.title) AS `movies for Tom Cruise`
```

"movies for Tom Cruise"
["Jerry Maguire", "Top Gun", "A Few Good Men"]

```
MATCH (actor:Person)-[:ACTED_IN]->(m:Movie)<-[:DIRECTED]-(director:Person)
RETURN actor.name, director.name, count(m) AS collaborations, collect(m.title) AS movies
```

"actor.name"	"director.name"	"collaborations"	"movies"
"Hugo Weaving"	"Lana Wachowski"	4	["The Matrix Reloaded", "The Matrix Revolutions", "Cloud Atlas", "The Matrix"]
"Laurence Fishburne"	"Lana Wachowski"	3	["The Matrix Reloaded", "The Matrix Revolutions", "The Matrix"]
"Carrie-Anne Moss"	"Lana Wachowski"	3	["The Matrix Reloaded", "The Matrix Revolutions", "The Matrix"]
"Keanu Reeves"	"Lana Wachowski"	3	["The Matrix Reloaded", "The Matrix Revolutions", "The Matrix"]
"Hugo Weaving"	"Andy Wachowski"	4	["The Matrix Reloaded", "The Matrix"]

Mittwoch, 29. Dezember 2021 11:12

```
MATCH (p:Person)-[:ACTED_IN]->(m:Movie)
WHERE p.name = 'Tom Cruise'
RETURN collect(m) AS `movies for Tom Cruise`
```



```
[
{
  "identity": 47,
  "labels": [
    "Movie"
  ],
  "properties": {
    "tagline": "The rest of his life begins now.",
    "title": "Jerry Maguire",
    "released": 2000
  }
},
{
  "identity": 25,
  "labels": [
    "Movie"
  ],
  "properties": {
    "tagline": "...",
    "title": "A Few Good Men",
    "released": 1992
  }
}
...
]
```

MATCH (a:Person)-[:ACTED_IN]->(m:Movie)

RETURN m.title, collect(a.name) as cast, count(*) as castSize

"m.title"	"cast"	"castSize"
"The Matrix Reloaded"	["Hugo Weaving", "Laurence Fishburne", "Carrie-Anne Moss", "Keanu Reeves"]	4
"The Matrix Revolutions"	["Hugo Weaving", "Laurence Fishburne", "Carrie-Anne Moss", "Keanu Reeves"]	4
"The Devil's Advocate"	["Al Pacino", "Charlize Theron", "Keanu Reeves"]	3
"A Few Good Men"	["James Marshall", "Kevin Pollak", "J.T. Walsh", "Aaron Sorkin", "Cuba Gooding Jr.", "Christopher Guest", "Noah Wyle", "Kiefer Sutherland", "Kevin Bacon", "Demi Moore", "Jack Nicholson", "Tom Cruise"]	12
"Top Gun"	["Val Kilmer", "Meg Ryan", "Tom Skerritt", "Kelly McGillis", "Tom Cruise", "Anthony Edwards"]	6
"Tomb Raider"	["Tara O'Sullivan", "Bonnie Hunt", "Tom Miko"]	3


```
MATCH (m:Movie)
WHERE m.title CONTAINS 'Matrix'
RETURN m{.title, .released} AS movie
```

"movie"
{"title":"The Matrix Reloaded","released":2003}
{"title":"The Matrix Revolutions","released":2003}
{"title":"The Matrix","released":1999}

```
MATCH (a:Person)-[:ACTED_IN]->(m:Movie)
WITH a, count(a) AS numMovies, apoc.coll.sort(collect(m.title))[0..3] as movies
WHERE numMovies = 5
RETURN a.name, numMovies, movies
ORDER BY a.name
```

"a.name"	"numMovies"	"movies"
"Hugo Weaving"	5	["Cloud Atlas", "The Matrix", "The Matrix Reloaded"]
"Jack Nicholson"	5	["A Few Good Men", "As Good as It Gets", "Hoffa"]
"Meg Ryan"	5	["Joe Versus the Volcano", "Sleepless in Seattle", "Top Gun"]

```
MATCH (a:Person)-[:ACTED_IN]->(m:Movie)
WITH a, count(a) AS numMovies, apoc.coll.sort(collect(m.title))[0..3] as movies
WHERE numMovies = 5
UNWIND movies as movie
RETURN a.name, numMovies, movie
```

"a.name"	"numMovies"	"movie"
"Hugo Weaving"	5	"Cloud Atlas"
"Hugo Weaving"	5	"The Matrix"
"Hugo Weaving"	5	"The Matrix Reloaded"
"Jack Nicholson"	5	"A Few Good Men"
"Jack Nicholson"	5	"As Good as It Gets"
"Jack Nicholson"	5	"Hoffa"
"Meg Ryan"	5	"Joe Versus the Volcano"
"Meg Ryan"	5	"Sleepless in Seattle"
"Meg Ryan"	5	"Top Gun"

Erzeugung von Knoten

```
CREATE (m:Movie:Action {title: 'Batman Begins'})
```

```
CREATE
```

```
(:Person {name: 'Michael Caine', born: 1933}),
```

```
(:Person {name: 'Liam Neeson', born: 1952}),
```

```
(:Person {name: 'Katie Holmes', born: 1978}),
```

```
(:Person {name: 'Benjamin Melniker', born: 1913})
```

Labels hinzufügen

```
MATCH (m:Movie)
```

```
WHERE m.title = 'Batman Begins'
```

```
SET m:Fantasy
```

```
RETURN labels(m)
```

Properties hinzufügen

```
MATCH (m:Movie)
```

```
WHERE m.title = 'Batman Begins'
```

```
SET m.released = 2005, m.lengthInMinutes = 140
```

```
RETURN m
```

Löschen von Knoten

```
MATCH (p:Person)
```

```
WHERE p.name = 'Jane Doe'
```

```
DELETE p
```

Löschen von Knoten eines Labels

```
match (p:Person) detach delete p
```

Löschen alle Knoten

```
match (n) detach delete n
```

Erzeugung von Kanten

```
MATCH (a:Person), (m:Movie)
WHERE
  a.name = 'Michael Caine' AND
  m.title = 'Batman Begins'
CREATE (a)-[:ACTED_IN]->(m)
```

```
MATCH (a:Person), (m:Movie), (p:Person)
WHERE a.name = 'Liam Neeson' AND
  m.title = 'Batman Begins' AND
  p.name = 'Benjamin Melniker'
CREATE (a)-[:ACTED_IN]->(m)<-[:PRODUCED]-(p)
```

Kanten mit Properties

```
MATCH (a:Person), (m:Movie)
WHERE a.name = 'Katie Holmes' AND m.title = 'Batman Begins'
CREATE (a)-[rel:ACTED_IN {roles: ['Rachel', 'Rachel Dawes']}]>(m)
```

Knoten und Kanten zusammen

```
MATCH (m:Movie)
WHERE m.title = 'Batman Begins'
CREATE (a:Person)-[:ACTED_IN]->(m)
SET a.name = 'Gary Oldman', a.born=1958
```

Löschen von Kanten

```
MATCH (a:Person)-[rel:WROTE | DIRECTED]->(m:Movie)
WHERE a.name = 'Katie Holmes' AND m.title = 'Batman Begins'
DELETE rel
```